
Charon
    

Charon is an extensible auth framework for Elixir, mostly for API's. The base package provides token generation & verification, and session handling. Additional functionality can be added with child packages. The package is opinionated with sane config defaults as much as possible.

  Features


	Striking a "statefulness" balance is left to the application. By default, access token are fully stateless, and refresh tokens are backed by a server-side session.
	A single base secret is used to derive special-purpose keys (like Phoenix's :secret_key_base), simplifying key management
	Tokens are generated by a factory behaviour with a default implementation using symmetric-key signed JWT's.
	Default token factory supports key rotation (including new derived keys)
	Default token factory supports asymmetric signatures using EdDSA with Ed25519 or Ed448 curves.
	Token verification is completely customizable using simple helper functions.
	Sessions are managed by a session store behaviour with a default implementation using Redis.
	To support both web- and other clients, it is possible to "split-off" the token's signatures and put them in cookies (to prevent storing tokens in LocalStorage and other insecure storage).
This idea is inspired by this excellent Medium article by Peter Locke.
	Flexible configuration that does not (have to) depend on the application environment.
	Small number of dependencies.


  
    
  
  Child packages


	CharonAbsinthe for using Charon with Absinthe.
	CharonOauth2 adds Oauth2 authorization server capability.
	CharonLogin adds authentication flows/challenges to generically provide login checks like password and MFA.
	(planned) CharonSocial for social login.
	(planned) CharonPermissions for authorization checks.


  
    
  
  Documentation


Documentation can be found at https://hexdocs.pm/charon.

  
    
  
  How to use



  
    
  
  Installation


The package can be installed by adding charon to your list of dependencies in mix.exs:
def deps do
  [
    {:charon, "~> 3.0"}
  ]
end

  
    
  
  Configuration


Configuration has been made easy using a config helper struct Charon.Config, which has a function from_enum/1 that verifies that your config is complete and valid, raising on missing fields. By using multiple config structs, you can support multiple configurations within a single application. The main reason to use multiple sets of configuration is that you can support different auth requirements in this way. You could, for example, create a never-expiring session for ordinary users and create a short-lived session for application admins.
# Charon itself only requires a token issuer and a base secret getter.
@my_config Charon.Config.from_enum(
             token_issuer: "MyApp",
             get_base_secret: &MyApp.get_base_secret/0
           )

# it is possible to use the application environment as well if you wish
@my_config Application.compile_env(:my_app, :charon) |> Charon.Config.from_enum()

  
    
  
  Setting up a session store


A session store can be created using multiple state stores, be it a database or a GenServer. All you have to do is implement a simple behaviour which you can find in Charon.SessionStore.Behaviour. Two default implementations are provided, Charon.SessionStore.RedisStore uses a Redis database, Charon.SessionStore.LocalStore uses a GenServer. Use Charon.SessionStore.DummyStore in case you don't want to use server-side sessions and prefer fully stateless tokens. The default and recommended option is RedisStore. Use LocalStore for local testing only - it is NOT persistent.
In order to use RedisStore, add Charon.SessionStore.RedisStore to your supervision tree:
# application.ex

def start(_, ) do
  redix_opts = [host: "localhost", port: 6379, password: "supersecret", database: 0]

  children = [
    ...
    {Charon.SessionStore.RedisStore, pool_size: 15, redix_opts: redix_opts},
    ...
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end

  
    
  
  Setting up a token factory


The default token factory is Charon.TokenFactory.Jwt, which requires no configuration by default, and uses symmetric HMAC-SHA256 signatures and a derived key. You can override its settings or set up your own TokenFactory by implementing Charon.TokenFactory.Behaviour.

  
    
  
  Protecting routes


Verifying incoming tokens is supported by the plugs in Charon.TokenPlugs (and submodules).
You can use these plugs to create pipelines to verify the tokens and their claims.
Note that the plugs don't halt the connection until you call Charon.TokenPlugs.verify_no_auth_error/2, but further processing stops as soon as a previous plug adds an error to the conn.
Example access- and refresh token pipelines:
defmodule MyApp.AccessTokenPipeline do
  @moduledoc """
  Verify access tokens. A access token:
   - must have a valid signature
   - must not be expired (and already valid)
   - must have a "type" claim with value "access"
  """
  use Plug.Builder
  import Charon.TokenPlugs

  @config Application.compile_env(:my_app, :charon) |> Charon.Config.from_enum()

  plug :get_token_from_auth_header
  plug :get_token_from_cookie, @config.access_cookie_name
  plug :verify_token_signature, @config
  plug :verify_token_nbf_claim
  plug :verify_token_exp_claim
  plug :verify_token_claim_equals, {"type", "access"}
  plug :verify_no_auth_error, &MyApp.TokenErrorHandler.on_error/2
  plug Charon.TokenPlugs.PutAssigns
end

defmodule MyApp.RefreshTokenPipeline do
  @moduledoc """
  Verify refresh tokens. A refresh token:
   - must have a valid signature
   - must not be expired (and already valid)
   - must have a "type" claim with value "refresh"
   - must have a corresponding session
   - must be fresh (see `Charon.TokenPlugs.verify_token_fresh/2` docs)
  """
  use Plug.Builder
  import Charon.TokenPlugs

  @config Application.compile_env(:my_app, :charon) |> Charon.Config.from_enum()

  plug :get_token_from_auth_header
  plug :get_token_from_cookie, @config.refresh_cookie_name
  plug :verify_token_signature, @config
  plug :verify_token_nbf_claim
  plug :verify_token_exp_claim
  plug :verify_token_claim_equals, {"type", "refresh"}
  plug :load_session, @config
  plug :verify_token_fresh, 10
  plug :verify_no_auth_error, &MyApp.TokenErrorHandler.on_error/2
  plug Charon.TokenPlugs.PutAssigns
end

# use the pipelines in your router
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  pipeline :api do
    plug :accepts, ["json"]
  end

  pipeline :valid_access_token do
    # or just use the token plugs right here instead of putting them in a separate module
    plug MyApp.AccessTokenPipeline
  end

  pipeline :valid_refresh_token do
    plug MyApp.RefreshTokenPipeline
  end

  scope "/" do
    pipe_through [:api]

    post "/current_session", SessionController, :login
  end

  scope "/" do
    pipe_through [:api ,:valid_access_token]

    delete "/current_session", SessionController, :logout
  end

  scope "/" do
    pipe_through [:api ,:valid_refresh_token]

    # optionally limit the refresh cookie path to this path using `Config.refresh_cookie_opts`
    post "/current_session/refresh", SessionController, :refresh
  end
end

  
    
  
  Logging in, logging out and refreshing


Create a session controller with login, logout and refresh routes. You can use Charon.SessionPlugs for all operations.
defmodule MyAppWeb.SessionController do
  @moduledoc """
  Controller for a user's session(s), including login, logout and refresh.
  """
  use MyAppWeb, :controller

  alias Charon.{SessionPlugs, Utils}
  alias MyApp.{User, Users}

  @config Application.compile_env(:my_app, :charon) |> Charon.Config.from_enum()

  @doc """
  Login to my app!
  """
  def login(conn, %{
        "email" => email,
        "password" => password,
        "token_transport" => token_transport
      })
      when token_transport in ~w(bearer cookie_only cookie) do
    with {:ok, user} <- Users.get_by(email: email) |> Users.verify_password(password) do
      # you can do extra checks here, like checking if the user is active, for example

      conn
      # you can add/override claims in the tokens (be careful!)
      |> SessionPlugs.upsert_session(
        @config,
        user_id: user.id,
        token_transport: token_transport,
        access_claim_overrides: %{"roles" => user.roles}
      )
      |> put_status(201)
      |> send_token_response(user)
    else
      _error -> send_resp(conn, 401, "user not found or wrong password")
    end
  end

  @doc """
  Logout from my app!
  """
  def logout(conn, _params) do
    conn
    |> SessionPlugs.delete_session(@config)
    |> send_resp(204, "")
  end

  @doc """
  Stay fresh with my app!
  """
  def refresh(%{assigns: %{current_user_id: user_id}} = conn, _params) do
    with %User{status: "active"} = user <- Users.get_by(id: user_id) do
      # here you can do extra checks again

      conn
      # there's no need to set user_id, token transport or extra session payload
      # because these are grabbed from the current session
      # but all added/overridden token claims must be passed in again
      |> SessionPlugs.upsert_session(@config, access_claim_overrides: %{"roles" => user.roles})
      |> send_token_response(user)
    else
      _error -> send_resp(conn, 401, "user not found or inactive")
    end
  end

  ###########
  # Private #
  ###########

  defp send_token_response(conn, user) do
    session = conn |> Utils.get_session() |> Map.from_struct()
    tokens = conn |> Utils.get_tokens() |> Map.from_struct()
    json(conn, %{tokens: tokens, session: session})
  end
end
And that's it :) Optionally, you can add get-all, logout-all and logout-other session endpoints, if your session store supports it (the default Redis one does).

  
    
  
  Testing


To aid in testing, some utility functions have been added for your convenience.
defmodule MyAppWeb.ControllerTest do
  use MyAppWeb.ConnCase
  alias Charon.TestHelpers
  alias MyApp.User

  setup seeds do
    # create a user somehow...
    user = %User{id: 1}

    # create a valid session with tokens and put the access token on the conn
    conn = TestHelpers.put_token(conn, user.id, MyAppWeb.Charon.config())

    [conn: conn]
  end

end

  
    
  
  Copyright and License


Copyright (c) 2023, YipYip B.V.
Charon source code is licensed under the Apache-2.0 License


  

    
LICENSE
    

                             Apache License
                       Version 2.0, January 2004
                    http://www.apache.org/licenses/
   TERMS AND CONDITIONS FOR USE, REPRODUCTION, AND DISTRIBUTION
	Definitions.
"License" shall mean the terms and conditions for use, reproduction,
and distribution as defined by Sections 1 through 9 of this document.
"Licensor" shall mean the copyright owner or entity authorized by
the copyright owner that is granting the License.
"Legal Entity" shall mean the union of the acting entity and all
other entities that control, are controlled by, or are under common
control with that entity. For the purposes of this definition,
"control" means (i) the power, direct or indirect, to cause the
direction or management of such entity, whether by contract or
otherwise, or (ii) ownership of fifty percent (50%) or more of the
outstanding shares, or (iii) beneficial ownership of such entity.
"You" (or "Your") shall mean an individual or Legal Entity
exercising permissions granted by this License.
"Source" form shall mean the preferred form for making modifications,
including but not limited to software source code, documentation
source, and configuration files.
"Object" form shall mean any form resulting from mechanical
transformation or translation of a Source form, including but
not limited to compiled object code, generated documentation,
and conversions to other media types.
"Work" shall mean the work of authorship, whether in Source or
Object form, made available under the License, as indicated by a
copyright notice that is included in or attached to the work
(an example is provided in the Appendix below).
"Derivative Works" shall mean any work, whether in Source or Object
form, that is based on (or derived from) the Work and for which the
editorial revisions, annotations, elaborations, or other modifications
represent, as a whole, an original work of authorship. For the purposes
of this License, Derivative Works shall not include works that remain
separable from, or merely link (or bind by name) to the interfaces of,
the Work and Derivative Works thereof.
"Contribution" shall mean any work of authorship, including
the original version of the Work and any modifications or additions
to that Work or Derivative Works thereof, that is intentionally
submitted to Licensor for inclusion in the Work by the copyright owner
or by an individual or Legal Entity authorized to submit on behalf of
the copyright owner. For the purposes of this definition, "submitted"
means any form of electronic, verbal, or written communication sent
to the Licensor or its representatives, including but not limited to
communication on electronic mailing lists, source code control systems,
and issue tracking systems that are managed by, or on behalf of, the
Licensor for the purpose of discussing and improving the Work, but
excluding communication that is conspicuously marked or otherwise
designated in writing by the copyright owner as "Not a Contribution."
"Contributor" shall mean Licensor and any individual or Legal Entity
on behalf of whom a Contribution has been received by Licensor and
subsequently incorporated within the Work.

	Grant of Copyright License. Subject to the terms and conditions of
this License, each Contributor hereby grants to You a perpetual,
worldwide, non-exclusive, no-charge, royalty-free, irrevocable
copyright license to reproduce, prepare Derivative Works of,
publicly display, publicly perform, sublicense, and distribute the
Work and such Derivative Works in Source or Object form.

	Grant of Patent License. Subject to the terms and conditions of
this License, each Contributor hereby grants to You a perpetual,
worldwide, non-exclusive, no-charge, royalty-free, irrevocable
(except as stated in this section) patent license to make, have made,
use, offer to sell, sell, import, and otherwise transfer the Work,
where such license applies only to those patent claims licensable
by such Contributor that are necessarily infringed by their
Contribution(s) alone or by combination of their Contribution(s)
with the Work to which such Contribution(s) was submitted. If You
institute patent litigation against any entity (including a
cross-claim or counterclaim in a lawsuit) alleging that the Work
or a Contribution incorporated within the Work constitutes direct
or contributory patent infringement, then any patent licenses
granted to You under this License for that Work shall terminate
as of the date such litigation is filed.

	Redistribution. You may reproduce and distribute copies of the
Work or Derivative Works thereof in any medium, with or without
modifications, and in Source or Object form, provided that You
meet the following conditions:
(a) You must give any other recipients of the Work or
Derivative Works a copy of this License; and
(b) You must cause any modified files to carry prominent notices
stating that You changed the files; and
(c) You must retain, in the Source form of any Derivative Works
that You distribute, all copyright, patent, trademark, and
attribution notices from the Source form of the Work,
excluding those notices that do not pertain to any part of
the Derivative Works; and
(d) If the Work includes a "NOTICE" text file as part of its
distribution, then any Derivative Works that You distribute must
include a readable copy of the attribution notices contained
within such NOTICE file, excluding those notices that do not
pertain to any part of the Derivative Works, in at least one
of the following places: within a NOTICE text file distributed
as part of the Derivative Works; within the Source form or
documentation, if provided along with the Derivative Works; or,
within a display generated by the Derivative Works, if and
wherever such third-party notices normally appear. The contents
of the NOTICE file are for informational purposes only and
do not modify the License. You may add Your own attribution
notices within Derivative Works that You distribute, alongside
or as an addendum to the NOTICE text from the Work, provided
that such additional attribution notices cannot be construed
as modifying the License.
You may add Your own copyright statement to Your modifications and
may provide additional or different license terms and conditions
for use, reproduction, or distribution of Your modifications, or
for any such Derivative Works as a whole, provided Your use,
reproduction, and distribution of the Work otherwise complies with
the conditions stated in this License.

	Submission of Contributions. Unless You explicitly state otherwise,
any Contribution intentionally submitted for inclusion in the Work
by You to the Licensor shall be under the terms and conditions of
this License, without any additional terms or conditions.
Notwithstanding the above, nothing herein shall supersede or modify
the terms of any separate license agreement you may have executed
with Licensor regarding such Contributions.

	Trademarks. This License does not grant permission to use the trade
names, trademarks, service marks, or product names of the Licensor,
except as required for reasonable and customary use in describing the
origin of the Work and reproducing the content of the NOTICE file.

	Disclaimer of Warranty. Unless required by applicable law or
agreed to in writing, Licensor provides the Work (and each
Contributor provides its Contributions) on an "AS IS" BASIS,
WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or
implied, including, without limitation, any warranties or conditions
of TITLE, NON-INFRINGEMENT, MERCHANTABILITY, or FITNESS FOR A
PARTICULAR PURPOSE. You are solely responsible for determining the
appropriateness of using or redistributing the Work and assume any
risks associated with Your exercise of permissions under this License.

	Limitation of Liability. In no event and under no legal theory,
whether in tort (including negligence), contract, or otherwise,
unless required by applicable law (such as deliberate and grossly
negligent acts) or agreed to in writing, shall any Contributor be
liable to You for damages, including any direct, indirect, special,
incidental, or consequential damages of any character arising as a
result of this License or out of the use or inability to use the
Work (including but not limited to damages for loss of goodwill,
work stoppage, computer failure or malfunction, or any and all
other commercial damages or losses), even if such Contributor
has been advised of the possibility of such damages.

	Accepting Warranty or Additional Liability. While redistributing
the Work or Derivative Works thereof, You may choose to offer,
and charge a fee for, acceptance of support, warranty, indemnity,
or other liability obligations and/or rights consistent with this
License. However, in accepting such obligations, You may act only
on Your own behalf and on Your sole responsibility, not on behalf
of any other Contributor, and only if You agree to indemnify,
defend, and hold each Contributor harmless for any liability
incurred by, or claims asserted against, such Contributor by reason
of your accepting any such warranty or additional liability.


   END OF TERMS AND CONDITIONS


  

    
Upgrade guide
    


  
    
  
  2.x to 3.x



  
    
  
  RedisStore changes


If you use Charon.SessionStore.RedisStore you must do the following before upgrading to 3.x:
	 Update Charon to >= 2.8.0
	 Upgrade your Redis instance to 7.x.x (this can be an in-place upgrade)
	 Run Charon.SessionStore.RedisStore.migrate_sessions/1
	 If you explicitly set config options :allow_unsigned? and :redix_module, you must remove your overrides. Support for these config options has been dropped.
	 Initialize the Redix connection pool in your supervision tree. See readme section setting up a sessionstore.
	 Remove your Redix module (if you don't use it for anything else).


  
    
  
  Deprecated functions


All marked-as-deprecated functions have been dropped, so make sure your code does not use these functions anymore. The compiler emits warnings if you do.


  

    
Changelog
    


  
    
  
  3.x



  
    
  
  Breaking


	Charon.SessionStore.RedisStore
	requires Redis >= 7.x.x
	uses a Redix connection pool by itself, which requires initialization under the application supervision tree
	implements optimistic locking
	uses a new storage format based on hash sets, to which sessions are migrated starting from Charon 2.8
	uses Redis functions to implement all session store operations in a single round trip to the Redis instance
	support for unsigned binaries has been dropped	config options :allow_unsigned? has been removed
	migrate_sessions/1 has been removed
	sessions that have not been migrated using migrate_sessions/1 can no longer be used




	Charon.SessionStore.LocalStore
	implements optimistic locking


	2.x marked-deprecated functions have been removed:
	Charon.Models.Session.deserialize/2
	Charon.Models.Session.serialize/1
	Charon.SessionStore.delete/3
	Charon.SessionStore.get/3
	Charon.SessionStore.delete_all/2
	Charon.SessionStore.get_all/2
	Charon.SessionStore.RedisStore.cleanup/1
	Charon.TokenPlugs.verify_refresh_token_fresh/2


	Charon.TokenPlugs.verify_token_signature/2 no longer adds default value "full" for claim "styp".
This should not result in issues for tokens created by Charon 2.x.x deployments.
Older deployments may wish to add a plug after Charon.TokenPlugs.verify_token_signature/2 that adds this default claim.




  

    
Charon 
    



      
Authentication & sessions for API's.

      




  

    
Charon.Config 
    



      
Config struct.
All config is read at runtime. So if, for example, you wish to override :session_ttl
based on the username, you can simply alter the config struct in your code.
That being said, config that HAS to be read at runtime, like secrets,
is stored as a getter to emphasize the fact and prevent you from accidentally setting
a compile-time value even if you put the config struct in a module attribute.
That is why the base secret has to be passed in via :get_base_secret.

  
    
  
  Keys & defaults


[
  :token_issuer,
  :get_base_secret,
  access_cookie_name: "_access_token_signature",
  access_cookie_opts: [http_only: true, same_site: "Strict", secure: true],
  # 15 minutes
  access_token_ttl: 15 * 60,
  enforce_browser_cookies: false,
  json_module: Jason,
  optional_modules: %{},
  refresh_cookie_name: "_refresh_token_signature",
  refresh_cookie_opts: [http_only: true, same_site: "Strict", secure: true],
  # 2 months
  refresh_token_ttl: 2 * 30 * 24 * 60 * 60,
  session_store_module: Charon.SessionStore.RedisStore,
  # 1 year
  session_ttl: 365 * 24 * 60 * 60,
  token_factory_module: Charon.TokenFactory.Jwt
]

  
    
  
  Glossary


	:access_cookie_name Name of the cookie in which the access token or its signature is stored.
	:access_cookie_opts Options passed to Plug.Conn.put_resp_cookie/3. Note that :max_age is set by Charon.SessionPlugs based on the token TTL. Overrides are merged into the defaults.
	:access_token_ttl Time in seconds until a new access token expires. This time may be reduced so that the token does not outlive its session.
	:enforce_browser_cookies If a browser client is detected, enforce that tokens are not returned to it as fully valid bearer tokens, but are transported (wholly or in part) as cookies.
	:get_base_secret Getter for Charon's base secret from which other keys are derived. Make sure it has large entropy (>= 256 bits). For example fn -> Application.get_env(:my_app, :charon_secret) end.
	:json_module The JSON module, like Jason or Poison.
	:optional_modules Configuration for optional modules, like Charon.TokenFactory.Jwt or CharonOauth2. See the optional module's docs for info on its configuration options.
	:refresh_cookie_name Name of the cookie in which the refresh token or its signature is stored.
	:refresh_cookie_opts Options passed to Plug.Conn.put_resp_cookie/3. Note that :max_age is set by Charon.SessionPlugs based on the token TTL. Overrides are merged into the defaults.
	:refresh_token_ttl Time in seconds until a new refresh token expires. This time may be reduced so that the token does not outlive its session.
	:session_store_module A module that implements Charon.SessionStore.Behaviour, used to store sessions.
	:session_ttl Time in seconds until a new session expires OR :infinite for non-expiring sessions.
	:token_factory_module A module that implements Charon.TokenFactory.Behaviour, used to create and verify authentication tokens.
	:token_issuer Value of the "iss" claim in tokens, for example "https://myapp.com"

    Charon.Models.Session - Charon v3.4.0
    
    

    


  
  

    
Charon.Models.Session 
    



      
A session.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Charon.Models.Tokens - Charon v3.4.0
    
    

    


  
  

    
Charon.Models.Tokens 
    



      
Tokens to be communicated to the client.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Charon.SessionPlugs - Charon v3.4.0
    
    

    


  
  

    
Charon.SessionPlugs 
    



      
Plugs to create, update/refresh and delete sessions.
When creating or updating a session, new tokens are created as well.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Charon.SessionStore - Charon v3.4.0
    
    

    


  
  

    
Charon.SessionStore 
    



      
The persistent session store. See Charon.SessionStore.Behaviour for callbacks.
All functions delegate to the configured session store module.
Client applications should use this module to interact with the underlying implementation,
because this module takes care of session struct version upgrades as well.

      




  

  
    
    Charon.SessionStore.Behaviour - Charon v3.4.0
    
    

    


  
  

    
Charon.SessionStore.Behaviour behaviour
    



      
Behaviour definition of a persistent session store.
Clients should not use the implementation directly, but should use Charon.SessionStore.

  
    
  
  Implementation guidelines


Implementations are expected to store sessions by ID, user ID and session type.
For the optional callbacks get_all/3 and delete_all/3, sessions should be retrievable
by user ID and session type only.
Implementations should return the session exactly as it went in
(they don't have to take care of struct version upgrades).
Implementations should handle cleanup of expired entries,
but may define additional functions and instructions to take care of such things
(like a cleanup/0 that should run periodically).
Implementations should implement optimistic locking with respect to the :lock_version field
of the session struct. On an update, the lock version of the updated session should match
the lock version of the existing session. If that is the case, the implementation should increase
:lock_version and proceed with the update. If not, the implementation should abort the update
and return {:error, :conflict}.

      


      
        Summary


  
    Callbacks
  


    
      
        
  
    
    Charon.SessionStore.DummyStore - Charon v3.4.0
    
    

    


  
  

    
Charon.SessionStore.DummyStore 
    



      
A dummy session store that can be used if fully stateless tokens are desired.

      




  

  
    
    Charon.SessionStore.LocalStore - Charon v3.4.0
    
    

    


  
  

    
Charon.SessionStore.LocalStore 
    



      
An in-memory persistent session store, implements behaviour Charon.SessionStore.
In addition to the required callbacks, this store also provides get_all/3 and delete_all/3 (for a user) functions.

  
    
  
  Usage


Add this store to your supervision tree to use it.
children = [
  Charon.SessionStore.LocalStore
]

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Charon.SessionStore.RedisStore - Charon v3.4.0
    
    

    


  
  

    
Charon.SessionStore.RedisStore 
    



      
A persistent session store based on Redis, which implements behaviour Charon.SessionStore.
In addition to the required callbacks, this store also provides get_all/3 and delete_all/3 (for a user) functions.

  
    
  
  Redis requirements


This module needs a Redis >= 7.0.0 instance and needs permissions to create Redis functions.
The optimistic-locking functionality of the store was not designed with a Redis cluster in mind
and will behave unpredictably when used with a distributed Redis deployment.
Using a failover-replica should be fine, however.
The Redis functions are registered with a name that includes the hashed Charon version,
to make sure that the called function matches the expected code,
and multiple Charon deployments can share a Redis instance.

  
    
  
  Config


Additional config is required for this module (see Charon.SessionStore.RedisStore.Config):
Charon.Config.from_enum(
  ...,
  optional_modules: %{
    Charon.SessionStore.RedisStore => %{
      key_prefix: "charon_",
      get_signing_key: &RedisStore.default_signing_key/1
    }
  }
)
The following options are supported:
	:key_prefix (optional). A string prefix for the Redis keys that are sessions.
	:get_signing_key (optional). A getter/1 that returns the key that is used to sign and verify serialized session binaries.


  
    
  
  Initialize store


RedisStore uses a connection pool and has to register Redis functions on startup.
In order to initialize it, you must add RedisStore to your application's supervision tree.
# in application.ex
def start(_, ) do
  redix_opts = [host: "localhost", port: 6379, password: "supersecret", database: 0]

  children = [
    ...
    {Charon.SessionStore.RedisStore, pool_size: 15, redix_opts: redix_opts},
    ...
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end

  
    
  
  Options


	:name (default module name) name of the supervisor
	:pool_size (default 10) total number of (local) normal workers
	:pool_max_overflow (default 5) max number of (local) extra workers in case of high load
	:redix_opts passed to Redix.start_link/1

The pool and its config options are local to the Elixir/OTP node,
so if you use multiple nodes, the total connection count to Redis maxes out at
(pool_size + pool_max_overflow) * number_of_nodes.

  
    
  
  Session signing


In order to offer some defense-in-depth against a compromised Redis server,
the serialized sessions are signed using a HMAC. The key is derived from the Charon
base secret, but can be overridden in the config. This is not usually necessary, except
when you're changing the base secret and still want to access your sessions.
It is debatable, of course, in case your Redis server is compromised,
if your application server holding the signing key can still be considered secure.
However, given that there are no real drawbacks to using signed session binaries,
since the performance cost is negligible, no extra config is needed, it is easy to implement,
and defense-in-depth is a good guiding principle, RedisStore implements this feature anyway.

  
    
  
  Implementation details


RedisStore stores sessions until their associated refresh token(s) expire.
That means that :refresh_expires_at is used to determine if a session is "alive",
not :expires_at.
This makes sense, because a session without a valid refresh token is effectively useless,
and it means we can support "infinite lifetime" sessions while still pruning sessions that aren't used.
This is one reason to set :refresh_ttl to a limited value, no more than six months is recommended.
Last but not least, Redis 7 functions
are used to implement some features, specifically optimistic locking, and to ensure all callbacks
use only a single round-trip to the Redis instance.

Charon.SessionStore.RedisStore.Config 
    



      
Config module for Charon.SessionStore.RedisStore.

Charon.TestHelpers 
    



      
Utility functions for writing tests.

Charon.TokenFactory 
    



      
Entrypoint for Charon.TokenFactory.Behaviour implementation.
All functions delegate to the configured module.

      




  

  
    
    Charon.TokenFactory.Behaviour - Charon v3.4.0
    
    

    


  
  

    
Charon.TokenFactory.Behaviour behaviour
    



      
Behaviour for token-signing modules.
Note that the token payload must be returned as a map with string keys on verification.
When the payload is serialized as JSON, this happens automatically.
However, when Erlang term format is used, this is not the case.

Charon.TokenFactory.Jwt 
    



      
JWT's with either symmetric (HMAC) or asymmetric (EDDSA) signatures.
The default, simplest and most performant option is symmetric signatures (MAC),
with the key derived from the Charon base secret.
Asymmetric tokens can be used when it is desirable for an external party
to be able to verify a token's integrity,
in which case distributing symmetric keys can be a hassle and a security risk.

  
    
  
  Keysets


In order to sign and verify JWT's, a keyset is used.
A keyset is a map of key ID's to keys.
A key is a tuple of the signing algorithm and the actual secret(s).
To simplify things and discourage key reuse,
a key can only be used with a single signing algorithm.
The default keyset looks like this, for example:
%{"default" => {:hmac_sha256, <<0, ...>>}}
Every token that is signed gets a "kid" claim in its header, allowing it to
be verified with the specific key and algorithm that it was signed with.

  
    
  
  Key cycling


It is possible to transition to a new signing key by adding a new key to the keyset
and setting it as the new signing key using the :signing_key config option:
%{
  "default" => {:hmac_sha256, <<0, ...>>},
  "new!" => {:hmac_sha512, <<1, ...>>}
}
Older tokens will be verified using the older key, based on their "kid" header claim.

  
    
  
  Tokens without a "kid" header claim


Legacy or external tokens may not have a "kid" header claim.
Such tokens can still be verified by adding
a "kid_not_set.<alg>" (for example "kid_not_set.HS256")
key to the keyset.

  
    
  
  Symmetric signatures


Symmetric signatures are message authentication codes or MACs,
either HMACs based on SHA256, 384 or 512,
or a MAC generated using Poly1305 or Blake3's keyed-hashing mode,
which can be used directly without using a HMAC wrapper.
Using Blake3 requires the optional dependency Blake3
and a key of exactly 256 bits.
By default, a SHA256-based HMAC is used.

  
    
  
  Asymmetric signatures


Asymmetric signatures are created using EDDSA (Edwards-curve Digital Signature Algorithm)
based on Curve25519 or Curve448.
Use in JWTs is standardized (pending) in RFC 8073.
These algorithms were chosen for performance and implementation ease,
since they offer built-in protection against many side-channel (timing) attacks and are
not susceptible to nonce-reuse (technically, they are, but not on the part of the implementation,
which means they are safe to use in your PlayStation).
Unless you are paranoid, use Curve25519, which offers about 128 bits of security.
Curve448 offers about 224 bits, but is significantly slower.
In order to use asymmetric signatures, generate a key using gen_keypair/1.
Create a publishable JWK using keypair_to_pub_jwk/1.
iex> keypair = Jwt.gen_keypair(:eddsa_ed25519)
iex> {:eddsa_ed25519, {_pubkey, _privkey}} = keypair
iex> %{"crv" => "Ed25519", "kty" => "OKP", "x" => <<_::binary>>} = Jwt.keypair_to_pub_jwk(keypair)

  
    
  
  Config


Additional config is required for this module (see Charon.TokenFactory.Jwt.Config):
Charon.Config.from_enum(
  ...,
  optional_modules: %{
    Charon.TokenFactory.Jwt => %{
      get_keyset: fn -> %{"key1" => {:hmac_sha256, "my_key"}} end,
      signing_key: "key1"
    }
  }
)
The following options are supported:
	:get_keyset (optional, default default_keyset/1). The keyset used to sign and verify JWTs. If not specified, a default keyset with a single key called "default" is used, which is derived from Charon's base secret.
	:signing_key (optional, default "default"). The ID of the key in the keyset that is used to sign new tokens.
	:gen_poly1305_nonce (optional, default :random). How to generate Poly1305-signed JWT nonces, can be overridden by a 0-arity function that must return a 96-bits binary. It is of critical importance that the nonce is unique for each JWT.


  
    
  
  Examples / doctests


# gracefully handles malformed tokens / unsupported algo's / invalid signature
iex> verify("a", @charon_config)
{:error, "malformed token"}
iex> verify("a.b.c", @charon_config)
{:error, "encoding invalid"}
iex> header = "notjson" |> url_encode()
iex> verify(header <> ".YQ.YQ", @charon_config)
{:error, "json invalid"}
iex> header = %{"missing" => "alg"} |> Jason.encode!() |> url_encode()
iex> verify(header <> ".YQ.YQ", @charon_config)
{:error, "malformed header"}
iex> header = %{"alg" => "boom"} |> Jason.encode!() |> url_encode()
iex> verify(header <> ".YQ.YQ", @charon_config)
{:error, "key not found"}
iex> header = %{"alg" => "HS256", "kid" => "default"} |> Jason.encode!() |> url_encode()
iex> verify(header <> ".YQ.YQ", @charon_config)
{:error, "signature invalid"}

# supports cycling to a new signing key, while still verifying old tokens
iex> {:ok, token} = sign(%{}, @charon_config)
iex> keyset = Jwt.default_keyset(@charon_config)
iex> keyset = Map.put(keyset, "ed25519_1", Jwt.gen_keypair(:eddsa_ed25519))
iex> config = override_opt_mod_conf(@charon_config, Jwt, get_keyset: fn _ -> keyset end, signing_key: "ed25519_1")
iex> {:ok, _} = verify(token, config)
iex> {:ok, new_token} = sign(%{}, config)
iex> new_token == token
false

# an old / external / legacy token without a "kid" claim can still be verified
# by adding a "kid_not_set.<alg>" key to the keyset
# a token MUST have an alg claim, which is mandatory according to the JWT spec
iex> [header, pl] = [%{"alg" => "HS256"}, %{}] |> Enum.map(&Jason.encode!/1) |> Enum.map(&url_encode/1)
iex> base = "#{header}.#{pl}"
iex> key = :crypto.strong_rand_bytes(32)
iex> signature = :crypto.mac(:hmac, :sha256, key, base) |> url_encode()
iex> token = "#{base}.#{signature}"
iex> {:error, "key not found"} = verify(token, @charon_config)
iex> keyset = %{"kid_not_set.HS256" => {:hmac_sha256, key}}
iex> config = override_opt_mod_conf(@charon_config, Jwt, get_keyset: fn _ -> keyset end)
iex> {:ok, _} = verify(token, config)

Charon.TokenFactory.Jwt.Config 
    



      
Config module for Charon.TokenFactory.Jwt.

Charon.TokenPlugs 
    



      
The plugs in this module (and its submodules) can be used to verify tokens.
The token's presence, signature, expiration and any claims can be checked.
Additionally, the token's session can be loaded and, in case of a refresh token,
it can be verified that it matches the session.
In case of validation errors, the plugs add an "auth error" to the conn,
but don't raise or halt the connection immediately.
This property can be used to support endpoints
that work with- or without authentication, for example,
or if you want to support multiple kinds of tokens.
The plug verify_no_auth_error/2 can be used to actually do something if there is an error.
All the plugs short-circuit, meaning that they immediately
return the connection if there are errors.
Using the plugs in these module, you can construct your own verification pipeline
using either Plug.Builder or standard Phoenix router pipelines.
Here are two examples for access- and refresh tokens, respectively,
that should be a good baseline for your own pipelines.

  
    
  
  Access tokens


defmodule MyApp.AccessTokenPipeline do
  use Plug.Builder

  @config Charon.Config.from_enum(Application.compile_env!(:my_app, :charon))

  plug :get_token_from_auth_header
  plug :get_token_from_cookie, @config.access_cookie_name
  plug :verify_token_signature, @config
  plug :verify_token_nbf_claim
  plug :verify_token_exp_claim
  plug :verify_token_claim_equals, {"type", "access"}
  plug :verify_no_auth_error, &MyApp.TokenErrorHandler.on_error/2
  plug Charon.TokenPlugs.PutAssigns
end

  
    
  
  Refresh tokens


defmodule MyApp.RefreshTokenPipeline do
  use Plug.Builder

  @config Charon.Config.from_enum(Application.compile_env!(:my_app, :charon))

  plug :get_token_from_auth_header
  plug :get_token_from_cookie, @config.refresh_cookie_name
  plug :verify_token_signature, @config
  plug :verify_token_nbf_claim
  plug :verify_token_exp_claim
  plug :verify_token_claim_equals, {"type", "refresh"}
  plug :load_session, @config
  plug :verify_token_fresh, 10
  plug :verify_no_auth_error, &MyApp.TokenErrorHandler.on_error/2
  plug Charon.TokenPlugs.PutAssigns
end

Charon.TokenPlugs.PutAssigns 
    



      
After verifying everything you would want to verify about a token,
assign the following to the conn:
	:user_id (bearer token claim "sub")
	:session_id (bearer token claim "sid")
	:token_payload
	:session (if fetched with Charon.TokenPlugs.load_session/2)

All of the assign names are overridable:
# assign the user ID to key :current_user_id
plug PutAssigns, claims: %{"sub" => :current_user_id, "sid" => :session_id}

  
    
  
  Doctests


iex> opts = PutAssigns.init([])
iex> conn = conn() |> set_token_payload(%{"sub" => 1, "sid" => "a"})
iex> conn |> PutAssigns.call(opts) |> Map.get(:assigns)
%{session_id: "a", token_payload: %{"sid" => "a", "sub" => 1}, user_id: 1}

iex> opts = PutAssigns.init(session: :da_session_baby)
iex> conn = conn() |> set_token_payload(%{"sub" => 1, "sid" => "a"}) |> set_session("hii")
iex> conn |> PutAssigns.call(opts) |> Map.get(:assigns)
%{
  session_id: "a",
  token_payload: %{"sid" => "a", "sub" => 1},
  user_id: 1,
  da_session_baby: "hii"
}

# skipped on auth error
iex> opts = PutAssigns.init([])
iex> conn = conn() |> set_token_payload(%{"sub" => 1, "sid" => "a"}) |> set_auth_error("boom")
iex> conn |> PutAssigns.call(opts) |> Map.get(:assigns)
%{}

Charon.Utils 
    



      
Utility functions, mainly getters and setters for module internals.

Charon.Utils.KeyGenerator 
    



      
Derive a key from a base secret using PBKDF2.

Charon.SessionPlugs.InsecureTokenTransportError exception
    



      
Error raised when token transport method "bearer" is requested by a browser client.

Charon.SessionPlugs.SessionStorageError exception
    



      
Error raised when a new/updated session could not be stored.

Charon.SessionPlugs.SessionUpdateConflictError exception
    



      
Error raised on an optimistic locking error when updating an existing session.

